

- *# Import some helpful packages for loading and plotting data*
- `using CSV, Dates, DataFrames, Gadfly, GLM, Statistics`

	Time	Replicate	OD
1	11:42:00	1	0.027
2	12:02:00	1	0.036
3	12:22:00	1	0.1
4	12:42:00	1	0.222
5	13:02:00	1	0.539
6	13:22:00	1	1.65
7	13:42:00	1	3.37
8	14:02:00	1	5.3
9	14:22:00	1	6.5
10	14:42:00	1	6.97
11	15:02:00	1	7.48
12	15:22:00	1	9.48

- `begin`
- *# Load CSV into a DataFrame*
- `csvfile = "Growth Curve Data.csv"`
- `df = DataFrame(CSV.File(csvfile))`
- `end`


	Time	Replicate	OD
1	0	1	0.027
2	20	1	0.036
3	40	1	0.1
4	60	1	0.222
5	80	1	0.539
6	100	1	1.65
7	120	1	3.37
8	140	1	5.3
9	160	1	6.5
10	180	1	6.97
11	200	1	7.48
12	220	1	9.48

```
• begin
•   # Normalise times and convert to minutes
•   start = df[1, :Time]
•   pdf = transform(df, :Time => ByRow(t -> Dates.value(Minute(t - start))) =>
•     :Time)
• end
```

Growth Curve Data


Given that OD is proportional to cell count (when properly diluted so that readings don't exceed 0.6), it can be used to track the growth of cells.

On a linear scale, this growth curve is an exponential, but be later made linear by applying a logarithmic transformation.



```
• # Construct a line-scatter plot, grouping by biological replicate  
• plot(pdf, x=:Time, y=:OD, color=:Replicate, Scale.color_discrete_hue,  
•     Guide.xlabel("Minutes"), Guide.ylabel("OD600"),  
•     Guide.title("<i>V. Natriegens</i> Growth Curves"))
```

A log transformation reveals that the region between 60 and 120 minutes can be safely said to be linear



```
• # Replot, but on a log-scale so that we can pick out the exponential growth region
• plot(pdf, x=:Time, y=:OD, color=:Replicate,
•     Scale.color_discrete_hue, Scale.y_log2,
•     Guide.xlabel("Minutes"), Guide.ylabel("OD600"),
•     Guide.title("<i>V. Natriegens</i> Growth Curves"))
```

logdf =

	Time	Replicate	OD
1	20	1	0.036
2	40	1	0.1
3	60	1	0.222
4	80	1	0.539
5	100	1	1.65

```
• # Trim the data to take a closer look at log-phase
• logdf = filter(:Time => t -> 20 <= t <= 100, pdf)
```

	Time	Replicate	OD
1	20	1	-4.79586
2	40	1	-3.32193
3	60	1	-2.17137
4	80	1	-0.891643
5	100	1	0.722466

- *# Log-transform the OD data*
- `transform!(logdf, :OD => ByRow(log2) => :OD)`


```
ols =  
StatsModels.TableRegressionModel{LinearModel{GLM.LmResp{Vector{Float64}}, GLM.DensePredCf  
OD ~ 1 + Time  
  
Coefficients:
```

	Coef.	Std. Error	t	Pr(> t)	Lower 95%	Upper 95%
(Intercept)	-6.13175	0.143444	-42.75	<1e-04	-6.58825	-5.67524
Time	0.0673347	0.0021625	31.14	<1e-04	0.0604526	0.0742167

- *# Perform an ordinary least-squares regression for a linear model*
- `ols = lm(@formula(OD ~ Time), logdf)`

```
[-4.78505, -3.43836, -2.09167, -0.744973, 0.601721]
```

- *# Insert a new column into our dataframe representing the model predictions*
- `logdf[!,:Model] = predict(ols)`



```

• # And then plot it on a log-scale
• plot(logdf, x=:Time, y=:OD, color=:Replicate,
•     Scale.color_discrete_hue, Geom.point,
•     Guide.xlabel("Minutes"), Guide.ylabel("OD600"),
•     Guide.title("<i>E. Coli</i> Growth Curves"),
•     # With the line-of-best-fit superimposed
•     layer(x=:Time, y=:Model, Geom.line))

```

Calculating Doubling-Time From Our Model

We can start with a fundamental equation that models the growth of microbes undergoing binary fission:

$$N = N_0 2^{\frac{t}{g}}$$

Where N is the current number of cells, N_0 is the initial number of cells, t is time, and g is generation or doubling-time. We want to rearrange this equation to fit the model $OD \sim \text{Time}$ after calculating the \log_2 of all ODs.

Let's start by applying the \log_2 to both sides of the equation:

$$\log_2 N = \log_2 N_0 + \frac{t}{g}$$

Ignoring the intercept and separating terms, we get an expression that matches our model:

$$\log_2 N = \frac{1}{g}t$$

Therefore we can conclude that g is equal to the reciprocal of our regression gradient.

The doubling time was ~14.9 minutes

```
• begin
•   # Calculate doubling-time
•   g = 1/coef(ols)[2]
•   # Format it into a nice string
•   md"The doubling time was ~$(round(g, sigdigits=3)) minutes"
• end
```