```
# Import some helpful packages for loading and plotting data
using CSV   , Dates   , DataFrames   , Gadfly   , GLM   , Statistics
```

	Time	Replicate	OD
1	10:30:00	1	0.028
2	10:30:00	2	0.027
3	10:30:00	3	0.026
4	10:30:00	4	0.027
5	10:50:00	1	0.041
6	10:50:00	2	0.043
7	10:50:00	3	0.052
8	10:50:00	4	0.04
9	11:10:00	1	0.082
10	11:10:00	2	0.084
	more		
60	15:10:00	4	4.66

```
begin
    # Load CSV into a DataFrame
    csvfile = "Growth Curve Data.csv"
    df = DataFrame(CSV.File(csvfile))
end
```

	Time	Replicate	OD
1	0	1	0.028
2	0	2	0.027
3	0	3	0.026
4	0	4	0.027
5	20	1	0.041
6	20	2	0.043
7	20	3	0.052
8	20	4	0.04
9	40	1	0.082
10	40	2	0.084
	more		
60	280	4	4.66

```
begin
    # Normalise times and convert to minutes
    start = df[1, :Time]
    pdf = transform(df, :Time => ByRow(t -> Dates.value(Minute(t - start))) => :Time)
end
```

Growth Curve Data

Given that OD is proportional to cell count (when properly diluted so that readings don't exceed 0.6), it can be used to track the growth of cells.

On a linear scale, this growth curve is an exponential, but be later made linear by applying a logarithmic transformation.




```
# Construct a line-scatter plot, grouping by biological replicate
plot(pdf, x=:Time, y=:OD, color=:Replicate, Scale.color_discrete_hue,
     Guide.xlabel("Minutes"), Guide.ylabel("OD₆₀₀"),
     Guide.title("<i>V. Natriegens</i> Growth Curves"))
```

A log transformation reveals that the region between `60` and `120` minutes can be safely said to be linear




```
# Replot, but on a log-scale so that we can pick out the exponential growth region
plot(pdf, x=:Time, y=:OD, color=:Replicate,
     Scale.color_discrete_hue, Scale.y_log2,
     Guide.xlabel("Minutes"), Guide.ylabel("OD₆₀₀"),
     Guide.title("<i>V. Natriegens</i> Growth Curves"))
```

logdf =

	Time	Replicate	OD
1	60	1	0.146
2	60	2	0.147
3	60	3	0.139
4	60	4	0.143
5	80	1	0.248
6	80	2	0.27
7	80	3	0.246
8	80	4	0.264
9	100	1	0.455
10	100	2	0.451
	more		
16	120	4	0.98

```
# Trim the data to take a closer look at log-phase
logdf = filter(:Time => t -> 60 <= t <= 120, pdf)
```

	Time	Replicate	OD
1	60	1	-2.77596
2	60	2	-2.76611
3	60	3	-2.84684
4	60	4	-2.80591
5	80	1	-2.01159
6	80	2	-1.88897
7	80	3	-2.02327
8	80	4	-1.92139
9	100	1	-1.13606
10	100	2	-1.1488
	more		
16	120	4	-0.0291463

```
# Log-transform the OD data
transform!(logdf, :OD => ByRow(log2) => :OD)
```

```
ols =
StatsModels.TableRegressionModel{LinearModel{GLM.LmResp{Vector{Float64}}, GLM.DensePredCho

OD ~ 1 + Time

Coefficients:
```

	Coef.	Std. Error	t	Pr(>|t|)	Lower 95%	Upper 95%
(Intercept)	-5.5174	0.0942555	-58.54	<1e-17	-5.71956	-5.31524
Time	0.0445808	0.00101638	43.86	<1e-15	0.0424009	0.0467607

```
# Perform and ordinary least-squares regression for a linear model
ols = lm(@formula(OD ~ Time), logdf)
```

```
[-2.84255, -2.84255, -2.84255, -2.84255, -1.95094, -1.95094, -1.95094, -1.95094, -1.05932,
```

```
# Insert a new column into our dataframe representing the model predictions
logdf[!,:Model] = predict(ols)
```




```
# And then plot it on a log-scale
plot(logdf, x=:Time, y=:OD, color=:Replicate,
     Scale.color_discrete_hue, Geom.point,
     Guide.xlabel("Minutes"), Guide.ylabel("OD₆₀₀"),
     Guide.title("<i>E. Coli</i> Growth Curves"),
     # With the line-of-best-fit superimpoosed
     layer(x=:Time, y=:Model, Geom.line))
```

Calculating Doubling-Time From Our Model

We can start with a fundamental equation that models the growth of microbes undergoing binary fission:

$$N = N_0 2^{\frac{t}{g}}$$

Where N is the current number of cells, N_0 is the initial number of cells, t is time, and g is generation or doubling-time. We want to rearrange this equation to fit the model `OD ~ Time` after calculating the $\log_2$ of all ODs.

Let's start by applying the $\log_2$ to both sides of the equation:

$$\log_2 N = \log_2 N_0 + \frac{t}{g}$$

Ignoring the intercept and separating terms, we get an expression that matches our model:

$$\log_2 N = \frac{1}{g} t$$

Therefore we can conclude that g is equal to the reciprocal of our regression gradient.

The doubling time was ~22.4 minutes

```
begin
    # Calculate doubling-time
    g = 1/coef(ols)[2]
    # Format it into a nice string
    md"The doubling time was ~$(round(g, sigdigits=3)) minutes"
end
```